

## Paper 305-2008

**Getting Started with ODS Statistical Graphics in SAS® 9.2**

Robert N. Rodriguez, SAS Institute Inc., Cary, NC

**ABSTRACT**

ODS Statistical Graphics (or ODS Graphics for short) is major new functionality for creating statistical graphics that is available in a number of SAS software products, including SAS/STAT®, SAS/ETS®, SAS/QC®, and SAS/GRAPH®. With the production release of ODS Graphics in SAS 9.2, over sixty statistical procedures have been modified to use this functionality, and they now produce graphs as automatically as they produce tables. In addition, new procedures in SAS/GRAPH use this functionality to produce plots for exploratory data analysis and for customized statistical displays. SAS/GRAPH is required for ODS Graphics functionality in SAS 9.2.

This paper presents the essential information you need to get started with ODS Graphics in SAS 9.2. ODS Graphics is an extension of ODS (the Output Delivery System), which manages procedure output and lets you display it in a variety of destinations, such as HTML and RTF. Consequently, many familiar features of ODS for tabular output apply equally to graphs. For statistical procedures that support ODS Graphics, you invoke this functionality with the `ods graphics on` statement. Graphs and tables created by these procedures are then integrated in your ODS output destination. ODS Graphics produces graphs in standard image file formats, and the consistent appearance and individual layout of these graphs are controlled by ODS styles and templates, respectively.

Since the default templates for procedure graphs are provided by SAS, you do not need to know the details of templates to create statistical graphics. However, with some understanding of the underlying Graph Template Language, you can modify the default templates to make changes to graphs that are permanently in effect each time you run the procedure. Alternatively, to facilitate making immediate changes to a particular graph, SAS 9.2 introduces the ODS Graphics Editor, a point-and-click interface with which you can customize titles, annotate points, and make other enhancements.

**INTRODUCTION**

Effective graphs are indispensable for modern statistical analysis. They reveal patterns, differences, and uncertainty that are not readily apparent in tabular output. Graphs provoke questions that stimulate deeper investigation, and they add visual clarity and rich content to reports and presentations.

In earlier SAS releases, creating graphs with statistical procedures usually required three additional programming steps. The first step was to create output data sets with the values to plot, including related statistical information such as *p*-values and sample sizes. The second step was to write a DATA step program that prepared these values for plotting. The third step was to render the plots with traditional SAS/GRAPH procedures and annotation.

In order to eliminate these steps and automate the creation of high-quality statistical graphics, SAS 9.1 introduced major new functionality, referred to as ODS Statistical Graphics (or ODS Graphics for short). Statistical procedures that have been modified to use this functionality now produce graphs as automatically as they produce tables, and these graphs are integrated with tables in ODS output. In SAS 9.1, two dozen procedures were modified to use ODS Graphics as experimental software; see Rodriguez (2004) and Rodriguez and Balan (2006). With the production release of ODS Graphics in SAS 9.2, this functionality is available with over sixty statistical procedures in SAS/STAT, SAS/ETS, SAS/QC, and Base SAS, and with new SAS/GRAPH procedures (see the list in “[STATISTICAL PROCEDURES THAT SUPPORT ODS GRAPHICS IN SAS 9.2](#)” on page 10). Note that SAS/GRAPH software is required for ODS Graphics functionality in SAS 9.2.

ODS Graphics is an extension of ODS (the Output Delivery System), which manages procedure output for display in a variety of destinations, such as HTML and RTF. Consequently, many ODS statements apply equally to tables and graphs, and you can build on your familiarity with ODS to get started with ODS Graphics.

ODS Graphics is enabled when you specify the following statement:

```
ods graphics on;
```

Statistical procedures that support ODS Graphics then create appropriate graphs, either by default or when you specify procedure options for requesting specific graphs. These options are documented in the “Syntax” section of the procedure chapter in the user’s guides for SAS/STAT, SAS/ETS, SAS/QC, and the statistical procedures in Base SAS; see SAS Institute Inc. (2008e, b, d, a). The “Details” section of the procedure chapter includes a subsection titled “ODS Graphics” that lists the available graphs, and many of these graphs are illustrated in the “Examples” section.

The following statements illustrate how you can create a default plot for a simple linear regression analysis:

```
ods graphics on;
ods html;


ods select ParameterEstimates FitPlot;
proc reg data=sashelp.Class;
  model Weight=Height;
quit;

ods html close;
ods graphics off;
```

You specify the `ods graphics on` statement to request ODS Graphics in addition to the usual tabular output produced by the REG procedure, and you specify the `ods html` statement to request the HTML destination. The `ods html close` statement closes the HTML destination, and the `ods graphics off` statement disables ODS Graphics. You do not need to disable ODS Graphics after each procedure step. Usually, you enable it once at the beginning of your SAS session, so that it stays enabled for the duration of the session. You should consider disabling ODS Graphics if your goal is solely to produce computational or tabular results, because ODS Graphics uses additional resources.

Figure 1 shows the output produced by the REG procedure for this example. The output consists of a table of parameter estimates and a fit plot, as requested with the `ods select` statement. Both the table and the plot are part of the default output (not shown here) produced by the REG procedure for a simple linear regression analysis; the default graphs also include regression diagnostics plots and a residuals plot. Note that the fit plot is accompanied by an inset that provides information relevant to the fit. This demonstrates how procedures that support ODS Graphics take advantage of computational results to enrich their graphs. With traditional graphics, creating a fit plot such as this one would require hundreds of lines of additional SAS program statements.

Figure 1 HTML Output with DEFAULT Style



The output in Figure 1 is displayed in the default ODS style for the HTML destination. ODS styles control the colors, fonts, and general appearance of all graphs and tables, and SAS 9.2 provides several styles that are recommended for use with statistical graphics. The following statements use the STATISTICAL style to produce the HTML output for the


regression example; [Figure 2](#) shows the output.

```
ods graphics on;
ods html style=statistical;

ods select ParameterEstimates FitPlot;
proc reg data=sashelp.Class;
  model Weight=Height;
quit;

ods html close;
ods graphics off;
```

**Figure 2** HTML Output with STATISTICAL Style



The following statements use the JOURNAL style to produce RTF output for the regression example; [Figure 3](#) shows the output.


```
ods graphics on;
ods rtf style=statistical;

ods select ParameterEstimates FitPlot;
proc reg data=sashelp.Class;
  model Weight=Height;
quit;

ods rtf close;
ods graphics off;
```

The JOURNAL style is a gray-scale style that is especially useful for graphs that will appear in journals and other black-and-white publications.

Figure 3 RTF Output with JOURNAL Style



The next section provides an overview of ODS Graphics, which explains the basics of creating and managing graphs. For a comprehensive introduction to ODS Graphics, see Chapter 21, “Statistical Graphics Using ODS,” in the *SAS/STAT 9.2 User’s Guide*. This is referred to as Chapter 21 throughout this paper.

## A PRIMER ON ODS STATISTICAL GRAPHICS

You invoke ODS Graphics by specifying the following statement:

```
ods graphics on;
```

ODS Graphics then remains in effect until you turn it off with the following statement:

```
ods graphics off;
```

As explained later in this paper, you use ODS GRAPHICS statement options to specify characteristics of your graphs, such as size, image format, and image file name. For details, see the “Syntax” section of Chapter 21.

Once you have invoked ODS Graphics, creating graphical output with procedures is as simple as creating tabular output. You can control your output in the following ways:

- ODS destination statements (such as ODS HTML or ODS RTF) specify where you want your graphs displayed. See the “ODS Destination Statements” section of Chapter 21 for a list of the supported destinations.
- Procedure options and defaults determine which graphs are created. For procedures that support ODS Graphics, these options are described in the “Syntax” section of the procedure chapters of the user’s guides for SAS/STAT, SAS/ETS, SAS/QC, and the statistical procedures in Base SAS. You usually request non-default graphs with the PLOTS= option in the procedure statement; the general behavior of this option is standard across procedures.

- ODS SELECT and ODS EXCLUDE statements select and exclude graphs from your output. As with tables, you refer to graphs by name in these statements. In the procedure chapters, the names of available graphs are listed in the “ODS Graphics” subsection of the “Details” section.
  - ODS OUTPUT statements create SAS data sets from the data objects used to make plots.
  - ODS styles control the general appearance and consistency of all graphs and tables; see “[GRAPH STYLES](#)” on page 5.
  - ODS templates modify the layout and details of each graph; see Chapter 21 for more information.
- NOTE:** A default template is provided by SAS for each graph, so you do not need to know anything about templates to create statistical graphics.

You can also access individual graphs, control the resolution and size of graphs, and modify your graphs as explained later in this paper.

## GRAPH STYLES

ODS styles control the overall appearance of graphs and tables. They specify colors, fonts, line styles, and other attributes of graph elements. The following styles are recommended for statistical work:

- The DEFAULT style is a color style intended for general-purpose work. See [Figure 1](#) for an example of this style, which is the default for the HTML destination.
- The STATISTICAL style is a color style recommended for output in Web pages or color print media. The STATISTICAL style might not necessarily print well on black-and-white devices. See [Figure 2](#) for an example. This is the default style for SAS/STAT documentation.
- The ANALYSIS style is a color style with a somewhat different appearance from the STATISTICAL style.
- The JOURNAL and JOURNAL2 styles are gray-scale and pure black-and-white styles, respectively, that are recommended for graphs that will appear in journals and in other black-and-white publications. See [Figure 3](#) for an example.
- The RTF style is used to produce graphs to insert into a Microsoft Word document or a Microsoft PowerPoint slide.

There are many other styles, including the LISTING style, which is the default for the LISTING destination.

You specify a style with the STYLE= option in the ODS destination statement. For example, the following statement requests HTML output produced with the JOURNAL style:

```
ods html style=Journal;
```

Similarly, the following statement sets the style for the LISTING destination:


```
ods listing style=Statistical;
```

Note that the style specified with the STYLE= option in the ODS LISTING statement applies only to graphs. The legacy SAS monospace format is used for tables.

## ODS DESTINATIONS

For most ODS destinations (including HTML, RTF, and PDF), graphs and tables are integrated in the output, and you view your output with an appropriate viewer, such as a web browser for HTML. However, the default LISTING destination is different. If you are using the LISTING destination in the SAS windowing environment, you view your graphs individually by clicking the graph icons in the Results window illustrated in [Figure 4](#). This action invokes a host-dependent graph viewer (for example, Microsoft Photo Editor on Windows). Note that graphs produced with ODS Graphics are *not* displayed with traditional graphs in the Graph window.

Figure 4 SAS Results Window



If you are using the SAS windowing environment and you prefer to view integrated output, you should specify a destination such as HTML or RTF. You can prevent the Output window from appearing by closing the LISTING destination, as in the following statements:

```
ods listing close;
ods html;
```

In general, a graph is created for every open destination. When you open a new destination, you should close all destinations that you do not need. This makes your jobs run faster and with fewer resources, because fewer graphs are produced.

### ACCESSING INDIVIDUAL GRAPHS

If you are writing a paper or creating a presentation, you need to access your graphs individually. There are various ways to do this, depending on the ODS destination. Three particularly useful methods are as follows:

- If you are viewing RTF output, you can simply copy and paste your graphs from the viewer into a Microsoft Word document or a Microsoft PowerPoint slide.
- If you are viewing HTML output, you can copy and paste your graphs from the viewer, or you can right-click the graph and save it to a file. Note that copying and pasting from RTF is preferable because the default resolution is higher than with HTML. See “SPECIFYING THE SIZE AND RESOLUTION OF GRAPHS” on page 7.
- You can save your graphs in image files and then include them into a paper or presentation. For example, you can save your graphs as PNG (portable network graphics) files and include them into a paper that you are writing with L<sup>A</sup>T<sub>E</sub>X or into an HTML document.

You can specify the graphics image format and the file name in the ODS GRAPHICS statement. For example, the following statements, when submitted before a procedure step that produces multiple graphs, save the graphs in PostScript files named *myname.ps*, *myname1.ps*, and so on:

```
ods listing close;
ods latex file="test.tex" path="C:\myfiles" gpath="C:\myfiles\ps";
ods graphics on / imagefmt=ps imagename="myname";
```

Chapter 21 provides details about the file types available with various destinations, how they are named, and how they are saved.

If you are using the LISTING destination and the SAS windowing environment, you can also copy your graphs from the default image viewer and paste into a Microsoft Word document or a Microsoft PowerPoint slide.

## SPECIFYING THE SIZE AND RESOLUTION OF GRAPHS

Two factors to consider when you are creating graphs for a paper or presentation are the size of your graph and its resolution. For best results, it is recommended that you specify the size of the graph as it will appear in the document (rather than resizing the graph after it has been produced). You can specify the size in the ODS GRAPHICS statement, as illustrated by the following examples:

```
ods graphics on / width=6in;
ods graphics on / height=4in;
ods graphics on / width=4.5in height=3.5in;
```

When only one dimension is specified, most graphs are produced with a default width/height aspect ratio of 4/3.

The default resolution of graphs created with the HTML and LISTING destinations is 100 DPI (dots per inch), whereas the default with the RTF destination is 200 DPI. You can change the resolution with the IMAGE\_DPI= option in any ODS destination statement, as in the following example:

```
ods html image_dpi=300;
```

An increase in resolution often improves the quality of the graphs, but it also increases the size of the image file. Chapter 21 provides more information about graph size and resolution.

## MODIFYING YOUR GRAPHS

Although ODS Graphics is designed to automate the creation of high-quality statistical graphics, you might occasionally need to modify your graphs. In SAS 9.2, there are two ways to proceed:

- You can use the ODS Graphics Editor, which provides a point-and-click interface, to make changes that are data-dependent and immediate. This approach is recommended if you are making ad hoc changes to a *specific* graph that you have created and are preparing for a paper or presentation.
- You can modify the ODS graph template for a plot to make changes that are persistent—in other words, applied *each time you run the procedure*.

The next two sections discuss these approaches.

## MODIFYING YOUR GRAPHS WITH THE ODS GRAPHICS EDITOR

You can use the ODS Graphics Editor to customize titles and labels, annotate data points, add text, and change graph element properties such as fonts, colors, and line styles. After you have modified your graph, you can save it as a PNG image file or as an SGE file, a special SAS file type which preserves the editing context. You can open previously saved SGE files with the Graph Editor and resume editing.

You can access the ODS Graphics Editor in the SAS windowing environment, provided that the LISTING destination is open and that you have enabled ODS Graphics to create editable graphs. There are two ways to enable editing:

- You can enable editing for the duration of your SAS session by first selecting the Results window and then entering `sgedit on` on the command line, as illustrated in [Figure 5](#). SAS confirms this by displaying the message “NOTE: Statistical Graphics Editor is ON.” at the bottom left, as illustrated in [Figure 6](#). Note that the command must be entered in the Results window.
- You can enable editing and make it the default setting across SAS sessions by changing the SAS Registry setting of ‘Statistical Graphics Editor’ to ‘On.’ To change this setting in the SAS windowing environment, first open the Registry Editor by typing `regedit` on the command line. Then select **SAS\_REGISTRY ► ODS ► GUI ► RESULTS**. Click on **Statistical Graphics Editor** to open the ‘Edit String Value’ window, and type **On** in the ‘Value Data’ line.

Creating editable graphs takes additional resources, so you might not want to permanently enable this feature. You can disable it within a SAS session by entering `sgedit off` on the command line. You can disable default editing across SAS sessions by changing the Registry setting of ‘Statistical Graphics Editor’ to ‘Off.’

Figure 5 Enabling Editing





Figure 6 Editing Enabled



To invoke the ODS Graphics Editor, submit your SAS program and then right-click in the Results window on the plot you want to edit and select **Edit**; see Figure 7.

Figure 7 Invoking the ODS Graphics Editor for an Editable Plot




Figure 8 shows the ODS Graphics Editor window for the editable diagnostic plot created by the ROBUSTREG procedure. In Figure 9, various tools in the ODS Graphics Editor have been used to modify the title and annotate a particular point. The edited plot can be saved as a PNG file or as a re-editable SGE file by selecting **File ► Save As**.

Figure 8 Diagnostic Plot before Editing





Figure 9 Diagnostic Plot after Editing



Note that the ODS Graphics Editor does not permit you to make structural changes to a graph (such as moving the positions of data points). The Editor provides you with a point-and-click way to make one-time changes to a specific graph, whereas modifying the graph template, as discussed in the next section, provides you with a programmatic way to make template changes that persist every time you run the procedure.

## MODIFYING YOUR GRAPHS BY EDITING GRAPH TEMPLATES

A graph template is a program, written in the Graph Template Language (GTL), that specifies the layout and details of a graph. Because SAS provides a default template for every graph produced by statistical procedures, you do not need to know anything about templates to create graphs with these procedures.

The GTL is a powerful language that includes statements for specifying plot layouts (such as lattices and overlays), plot types (such as scatter plots and histograms), and text elements (such as titles, footnotes, and insets). It also provides support for built-in computations (such as histogram binning and loess smoothing) and the evaluation of expressions. Options are available for specifying colors, marker symbols, and other attributes of plot features.

Graphs, like all SAS output, are constructed from two underlying components: a data object supplied by a procedure at run time and a compiled template that is designed to work with this data object. Together, the data object and the template form an output object that ODS displays in one or more output destinations.

The default graph templates provided by SAS are usually lengthy and complex because they specify a complete description of how the graph is to be produced by the procedure. To put it another way, graph templates accomplish and hide the work of producing a graph that formerly required post-processing of output data sets with user-written programs.

With moderate knowledge of the GTL, you can edit graph templates to make simple modifications such as changing titles and axis labels or adding footnotes with project information. “[EXAMPLE 2: STATISTICAL GRAPHICS FOR A SURVIVAL STUDY](#)” on page 14 illustrates modifications of this type. Chapter 21 provides information that can help you get started with this approach, including how to identify and access the template for a particular graph, basic GTL concepts, and examples of graph template modifications. For complete details of the GTL syntax, see *SAS/GRAPH: Graph Template Language Reference*.

Another reason for learning the GTL is that you can use it to create highly customized displays by writing your own graph templates and applying them directly to data with the SGRENDER procedure. See “[NEW STATISTICAL GRAPHICS PROCEDURES IN SAS/GRAPH](#)” on page 11.

## STATISTICAL PROCEDURES THAT SUPPORT ODS GRAPHICS IN SAS 9.2

The following statistical procedures have been enhanced to support ODS Graphics in SAS 9.2:

Base SAS	SAS/STAT		SAS/QC	SAS/ETS
CORR	ANOVA	MI	ANOM	ARIMA
FREQ	BOXPLOT	MIXED	CAPABILITY	AUTOREG
UNIVARIATE	CALIS	MULTTEST	CUSUM	ENTROPY
	CLUSTER	NPAR1WAY	MACONTROL	EXPAND
	CORRESP	PHREG	PARETO	MODEL
	FACTOR	PLS	RELIABILITY	PANEL
	FREQ	PRINCOMP	SHEWHART	RISK
	GAM	PRINQUAL		SIMILARITY
	GENMOD	PROBIT		SYSLIN
	GLIMMIX	QUANTREG		TIMESERIES
	GLM	REG		UCM
	GLMSELECT	ROBUSTREG		VARMAX
	KDE	RSREG		X12
	KRIGE2D	SEQDESIGN		
	LIFEREG	SEQTEST		
	LIFETEST	SIM2D		
	LOESS	TCALIS		
	LOGISTIC	TRANSREG		
	MCMC	TTEST		
	MDS	VARIOGRAM		

Support for ODS Graphics is experimental for SAS/QC procedures and for the UNIVARIATE and BOXPLOT procedures in SAS 9.2. For details about the specific graphs available with a particular procedure, see the “Syntax” and “ODS Graphics” sections of the procedure chapters in the user’s guides for SAS/STAT, SAS/ETS, SAS/QC, and the statistical procedures in Base SAS.

## PROCEDURES THAT SUPPORT ODS GRAPHICS AND TRADITIONAL GRAPHICS

A number of procedures that support ODS Graphics in SAS 9.2 also produce traditional graphics in previous releases of SAS. These include the UNIVARIATE procedure in Base SAS; the BOXPLOT, LIFEREG, LIFETEST, and REG procedures in SAS/STAT; and the ANOM, CAPABILITY, CUSUM, MACONTROL, PARETO, RELIABILITY, and SHEWHART procedures in SAS/QC. All of these procedures continue to produce traditional graphics, but in some cases, they do so only when ODS Graphics is not enabled. For more information about the interaction between traditional graphics and ODS graphics in these procedures, see the chapters for these procedures in their respective user’s guides.

Note that traditional graphs are saved in SAS graphics catalogs and are controlled by the GOPTIONS statement. In contrast, ODS Graphics produces graphs in standard image file formats (not graphics catalogs), and the appearance and layout of these graphs are controlled by ODS styles and templates, respectively.

## NEW STATISTICAL GRAPHICS PROCEDURES IN SAS/GRAPH

Statistical procedures that support ODS Graphics create graphs in the context of a specific analysis. There are many other contexts in which the use of statistical graphics plays a valuable role, including the exploration and preliminary examination of data and the construction of specialized displays for novel analyses. These situations require versatile, general-purpose graphical tools for the creation of standalone plots.

SAS 9.2 introduces a family of statistical graphics procedures in SAS/GRAPH that are designed to meet these needs. The following procedures use ODS Graphics functionality and provide a convenient syntax for creating a variety of plots directly from data:

- SGSCATTER creates single-cell and multi-cell scatter plots and scatter plot matrices with optional fits and ellipses.
- SGPLOT creates single-cell plots with a variety of plot and chart types.
- SGPANEL creates single-page or multi-page panels of plots and charts conditional on classification variables.

These procedures, which are collectively referred to as the “SG procedures,” can produce density plots, dot plots, needle plots, series plots, horizontal and vertical bar charts, histograms, and box plots. They can also compute and display loess fits, polynomial fits, penalized B-spline fits, reference lines, bands, and ellipses. Graphs produced with the SG procedures and statistical procedures have a consistent appearance that is determined by the ODS style. The SG procedures are documented in the *SAS/GRAPH Statistical Graphics Procedures Guide*.

For situations that require highly customized displays which are not available with the SG procedures, you can write your own graph templates, taking advantage of the power of the Graph Template Language. You can then apply these templates to your data and render the graphs with the SGRENDER procedure, which is also new in SAS/GRAPH. This use of the Graph Template Language is outside the scope of this paper, but Cartier (2006) provides a tutorial introduction and Chapter 21 provides an overview and examples. For complete documentation of the Graph Template Language, see the *SAS/GRAPH Graph Template Language Reference* (SAS Institute Inc. 2008c).

Note that you do not need to enable ODS Graphics in order to use the SG procedures. However, the options available in the ODS GRAPHICS statement are applicable to these procedures.

## EXAMPLE 1: STATISTICAL GRAPHICS FOR A LINEAR MODEL ANALYSIS

This example illustrates the start-to-finish use of ODS Graphics in an analysis in which the SGPANEL procedure in SAS/GRAPH creates a preliminary display of the data and the GLM procedure creates a specialized display that adds information to the statistical analysis.

The following statements create a data set that contains a response variable  $y$  and two classification variables,  $a$  and  $b$ .

```

data measure;
  drop i abEffect;
  do a = 1 to 3;
    do b = 1 to 3;
      if ((a = 3) & (b = 3)) then abEffect = 3;
      else abEffect = 1;
      do i = 1 to 10;
        y = abEffect + rannor(1);
        output;
      end;
    end;
  end;
run;
proc sort data=measure; by b;
run;

```

The next statements use the SGPANEL procedure to plot the means of  $y$  for levels of  $a$  in a display that is paneled by the levels of  $b$ . This is often referred to as a "means plot" or a "two-factor interaction plot."

```


ods html style=statistical;

title "Two-Factor Interaction Plot";
proc sgpanel data=measure;
  panelby b / columns=3 spacing=5;
  vline a / response=y stat=mean limits=both
  markers
  legendlabel= "Cell Means with 95% Confidence Limits";
  discretelegend;
run;
title;

```

The display, shown in [Figure 10](#), suggests the presence of an interaction effect, which should be included in a follow-up analysis of variance.

**Figure 10** Interaction Plot Produced with the SGPANEL Procedure



The next statements carry out the analysis of variance with the GLM procedure. The LSMEANS statement requests

least squares means (LS-means) for the interaction of a and b. The option PDIFF=ALL requests  $p$ -values for all pairwise differences of the LS-means.

```
ods graphics on;
ods select ModelANOVA DiffPlot;
proc glm data=measure;
  class a b;
  model y = a|b / ss3;
  lsmeans a*b / pdiff=all;
run;
ods graphics off;
ods html close;
```


The ANOVA table, shown in Figure 11, indicates that the main and interaction effects are significant.

**Figure 11** ANOVA Results from the GLM Procedure

The GLM Procedure					
Dependent Variable: y					
Source	DF	Type III SS	Mean Square	F Value	Pr > F
a	2	11.19595624	5.59797812	6.42	0.0026
b	2	18.09492619	9.04746309	10.37	<.0001
a*b	4	14.27143232	3.56785808	4.09	0.0045

When you specify an LSMEANS statement with the PDIFF= option, PROC GLM produces a default plot appropriate for the LS-means comparison. For PDIFF=ALL, the procedure produces a diffogram, which displays all pairwise LS-means differences and indicates which are significant. The diffogram, shown in Figure 12, gives you more information about the interaction by telling you which of the cell mean differences are significant.

**Figure 12** LS-Means Diffogram Produced with the GLM Procedure



This display is related to the mean-mean scatter plot of Hsu and Peruggia (1994); also see Hsu (1996). Diffograms are also available with the GLIMMIX procedure in SAS 9.2. For more details and examples, see the chapter on the GLIMMIX procedure in the *SAS/STAT User's Guide*.

## EXAMPLE 2: STATISTICAL GRAPHICS FOR A SURVIVAL STUDY

This example illustrates ODS Graphics by using a combination of SG procedures and statistical procedures in the context of a survival study. The UIS data set used here is a subset of data from the University of Massachusetts/Amherst Aids Research Unit (UMARU) IMPACT Study (UIS). This study consisted of two concurrent randomized trials of residential treatment for drug abuse, and the purpose was to compare treatment programs of different planned durations designed to reduce drug abuse. For additional background, see Hosmer and Lemeshow (1999).

The following statements create the UIS data set, which contains 628 observations and 12 variables. The response variable time is the survival time in days to return to drug use (measured from admission) of a participant. The censoring indicator variable censor has a value of 1 if a participant returned to drug use, and a value of 0 otherwise. For each participant, the variable treat indicates the treatment randomization assignment (0=Short and 1=Long), and the variable site indicates the treatment site (0=A and 1=B). There were 444 participants at Site A who received either a 3-month (short) or 6-month (long) treatment. There were 184 participants at Site B who received either a 6-month (short) or 12-month (long) treatment.

```
data UIS;
  input id age becktota hercoc ivhx ndrughtx race treat site lot time censor;
  label censor = "Returned to Drug Use"
         time = "Time to Return to Drug Use (Days)"
         treat = "Treatment"
         site = "Treatment Site";
  if ivhx=2 then ivhx=1;
  ndrughtx1 = 1/((ndrughtx+1)/10);
  ndrughtx2 = 1/((ndrughtx+1)/10) * log((ndrughtx+1)/10);
datalines;
1  39  9  4  3  1  0  1  0  123 188 1
2  33 34  4  2  8  0  1  0  25  26 1
... more lines ...
;
```

The following statements group the data so that you can examine the times by combinations of treatment and site. The variable EventType indicates whether a participant returned to drug use.

```
data grouped;
  set UIS;
  length EventType $ 9;
  label EventType="Type";
  if (treat=0 and site=0) then do; Group="Short*Site A"; Treatment="Short"; end;
  else if (treat=0 and site=1) then do; Group="Short*Site B"; Treatment="Short"; end;
  else if (treat=1 and site=0) then do; Group="Long*Site A"; Treatment="Long"; end;
  else do; Group="Long*Site B"; Treatment="Long"; end;
  if (censor=0) then EventType = 'Censored';
  else EventType = 'Event';
run;
```

The next statements use the SGPLOT procedure to plot the survival times for all combinations of treatment and site.

```
ods listing style=statistical;
proc sgplot data=grouped;
  title "Survivorship Data for Long and Short Treatments";
  footnote justify=left italic "Source: Applied Survival Analysis, Hosmer and Lemeshow";
  xaxis grid label = "Time to Return to Drug Use (Days)";
  yaxis label = " ";
  scatter x=time y=Group / group=EventType markerattrs=(symbol=circle) transparency=0.5;
run;

title;
footnote;
```

The SGPLOT procedure plots time for the four combinations of treatment and site, as shown in Figure 13. Note that the `ods graphics on` statement is not required with the SG procedures. Values of Group are displayed along the y-axis, and values of Time are displayed along the x-axis. The `MARKERATTRS=` option specifies the plotting symbol (a circle), and the `GROUP=` option differentiates the circles with a different color for each level of Eventtype. The actual colors (blue for Event and red for Censored) are determined by the `STATISTICAL` style. The `TRANSPARENCY=` option controls the density of the circles by specifying their degree of transparency.

**Figure 13** Display of Survival Times Produced with the SGPLOT Procedure




Figure 13 shows that the longer survival times are mostly censored. Specialized methods are needed to analyze the data. The product-limit (Kaplan-Meier) estimator is commonly used to describe the survivor function. The following statements use the LIFETEST procedure to create a plot of two product-limit functions for each site (one for each treatment program). First, the data are sorted by Site so that PROC LIFETEST can create survival curves for different sites in separate plots.


```
proc sort data=UIS;
  by site;
run;

ods graphics on;
proc lifetest data=grouped plots=survival(cb=hw test atrisk=0 to 1500 by 250);
  ods select SurvivalPlot;
  time time * censor(0);
  strata Treatment;
  by site;
run;
```

The `ods graphics on` statement is required for PROC LIFETEST to use ODS Graphics to produce plots. The `PLOTS=` option requests a graph of product-limit curves with 95% Hall-Wellner confidence bands, in addition to the number of subjects at risk at times specified with the `ATRISK=` option. Figure 14 shows the graph of the estimated survivor functions for Site A, which reveals that the long program participants survive longer than the short program participants. In other words, participants in the short program return to drug use earlier than those in the long program. The numbers of subjects at risk shown in Figure 14 are consistent with this conclusion. The  $p$ -value for the log-rank test ( $p=0.0047$ )

indicates a significant difference in the survivorship between the long and short programs. Similar conclusions can be drawn from the display for Site B (not shown here).

**Figure 14** Survival Plot for Site A Produced with the LIFETEST Procedure



The LIFETEST procedure, like other statistical procedures that use ODS Graphics, provide a PLOTS= option for modifying its graphs. When these options are not sufficient, you can modify the template for a graph to make changes that persist each time you run the procedure, as discussed in Chapter 21.

The following statements run PROC LIFETEST to determine the template name for the survival plot:

```
ods trace output;
proc lifetest data=grouped plots=survival(cb=hw test atrisk=0 to 1500 by 250);
ods select SurvivalPlot;
time time * censor(0);
strata Treatment;
by site;
run;
ods trace off;
```

The trace output results (not shown here) show that the template name is Stat.Lifetest.Graphics.ProductLimitSurvival. The following statements display the template (not shown here):

```
proc template;
source Stat.Lifetest.Graphics.ProductLimitSurvival;
run;
```

The next statements modify the template by changing the default title of the survival plot from “Product-Limit Survival Estimate” to “Kaplan-Meier Plot.” Complete details are provided in Example 21.6, “Customizing Survival Plots,” of Chapter 21, which also covers general aspects of managing templates that have been edited.

```
proc template;
define statgraph Stat.Lifetest.Graphics.ProductLimitSurvival;
dynamic NStrata xName plotAtRisk plotCensored plotCL plotHW plotEP
labelCL labelHW labelEP maxTime StratumID classAtRisk plotBand
plotTest GroupName yMin Transparency SecondTitle TestName pValue;
```


```

BeginGraph;
  if (NSTRATA=1)
    if (EXISTS(STRATUMID))
      entrytitle "Kaplan-Meier Plot for " STRATUMID;
    else
      entrytitle "Kaplan-Meier Plot";
... more lines ...
run;
proc lifetest data=grouped plots=survival(cb=hw test atrisk=0 to 1500 by 250);
ods select SurvivalPlot;
time time * censor(0);
strata Treatment;
by site;
run;

```

The modified survival plot is shown in [Figure 15](#). This particular modification can also be made with the ODS Graphics Editor, which is recommended for making changes to a specific graph as opposed to making persistent changes.

**Figure 15** Modified Survival Plot Produced with the LIFETEST Procedure



The following statements delete the modified template from SASUSER.TEMPLAT and revert to the default template in SASHELP.TMPLMST, which is where SAS always finds the templates it provides. Chapter 21 describes how you can use the ODS PATH statement with template libraries that you permanently create.

```

proc template;
  delete Stat.Lifetest.Graphics.ProductLimitSurvival;
run;

```

In order to account for the risk factors in the UIS data, the following statements fit a proportional hazards model and create an output data set that contains ten weighted Schoenfeld residual variables, W1-W10, one for each effect.

```

proc phreg data=grouped noprint;
  class ivhx(ref="1");
  model time * censor(0) = age becktota ndruggfp1 ndruggfp2 ivhx race treat
    site age*site race*site;
  output out=out1 wtressch=w1-w10;
run;

```

The output from the PHREG procedure is not shown here. You can evaluate the adequacy of the model by plotting the weighted Schoenfeld residuals against the survival times for each independent variable. Since the survival times are skewed, it is helpful to apply a log transformation to time before making the plots. This is done with the following statements:

```
data plot;
  set out1;
  log_time = log(time);
  label log_time = "log(Time)";
run;
```

The following statements use the SGSCATTER procedure to create the scatter plots. The PLOT statement specifies weighted residuals that correspond to the covariates Race, Treatment, Sites, and Race\*Site. The ROWS= and COLUMNS= options specify the layout for paneling the plots. The LOESS option smooths the residuals, and the GRID option adds a grid.


```
proc sgscatter data=plot;
  title 'Weighted Schoenfeld Residuals and Loess Smooths';
  label w1 = "age"      w2 = "becktota"
        w3 = "ndrugfp1" w4 = "ndrugfp2"
        w5 = "ivhx3"   w6 = "race"
        w7 = "treatment" w8 = "site"
        w9 = "agesite" w10 = "race * site";

  plot (w6-w8 w10)*log_time / rows    = 2
                                columns = 2
                                loess
                                grid
                                markerattrs = (size=5);

run;
title;
```

The residual display is shown in [Figure 16](#).

**Figure 16** Residual Display Produced with the SGSCATTER Procedure



Under the time-varying coefficients model, a plot of the weighted Schoenfeld residuals and their loess smooth should show no trend over time if the corresponding covariate has a proportional hazard. Three of the plots in [Figure 16](#) support this assumption. The plot for Treatment indicates that the effect of the longer treatment is more pronounced in the earlier and later periods of follow-up. In fact, Hosmer and Lemeshow (1999) concluded that this departure is not significant because the Wald test of the treat by log\_time interaction is not significant.

You might want to use the PHREG procedure to generate predicted survival curves for different sets of covariate values. You can specify the covariate values with the COVARIATES= data set in the BASELINE statement, and you can use the PLOTS= option in the PROC PHREG statement to display a survival curve for each row of covariates in this data set. Suppose you want to compare the baseline survival curves for a 25-year-old participant with the short treatment and a 50-year-old participant with the long treatment. The following statements create a data set named Row in which AGE=25 and TREAT=0 for the first row, and AGE=50 and TREAT=1 for the second row. The other covariates have identical values for the two rows. The variable Age\_Treat identifies the covariate sets.


```
data Row;
  length age_treat $ 27;
  label age_treat = "Age-Treatment Combination";
  input age becktota ndruggfp1 ndruggfp2 ivhx race treat site age_treat $27-56;
datalines;
25 17 2.5 -2.2907 1 1 0 0 Age 25 with Short Treatment
50 17 2.5 -2.2907 1 1 1 0 Age 50 with Long Treatment
;
run;
```

The following statements produce a plot of the survival curves, which is shown in [Figure 17](#).

```
proc phreg data=uis plots(overlay=stratum)=survival;
  ods select SurvivalPlot;
  class ivhx(ref="1");
  model time * censor(0) = age becktota ndruggfp1 ndruggfp2 ivhx race treat
    site age*site race*site;
  output out=out1 wtressch=w1-w10;
  baseline out=pred covariates=row survival=_all_ / rowid=age_treat;
run;
```

The PLOTS= option requests the survival plot, and the OVERLAY= option overlays the two curves in the same plot. The ROWID= option specifies that the values of Age\_Treat are to be used to identify the curves. The OUT=PRED and SURVIVAL=\_ALL\_ options create an output data set that contains the survival function estimates.

**Figure 17** Survival Plot Produced with the PHREG Procedure



The plot reveals that a 25-year-old participant in the short program is more likely to return to drug use than a 50-year-old in the long treatment program.

## CONCLUSIONS

The following table summarizes ways in which you can use ODS Graphics in SAS 9.2.

Task	What Do You Use?	What Is Involved?
Creating graphs for statistical analyses	Statistical procedures that support ODS Graphics	<code>ods graphics on</code> statement; graphs created by default or with procedure options
Creating standalone graphs for exploration of data or for customized displays	SGPLOT, SGPANEL, SGSCATTER procedures in SAS/GRAPH	Procedure syntax
Changing overall appearance of graphs and tables	ODS styles	STYLE= option in ODS destination statement
Enhancing specific graphs for presentation or paper	ODS Graphics Editor	Request editable graphs, invoke Editor, then use point-and-click interface
Making persistent, programmatic changes in graphs	SAS-supplied default ODS graph template	Modify default graph template with Graph Template Language and compile with TEMPLATE procedure
Creating highly customized graphs (outside scope of this paper)	User-written graph template	Write template with Graph Template Language, compile with TEMPLATE procedure, then apply to data with SGRENDER procedure in SAS/GRAPH

## REFERENCES

- Cartier, J. (2006), "A Programmer's Introduction to the Graphics Template Language," in *Proceedings of the Thirty-first Annual SAS Users Group International*, Cary, NC: SAS Institute Inc.
- Hosmer, D. W. and Lemeshow, S. (1999), *Applied Survival Analysis*, New York: John Wiley & Sons.
- Hsu, J. C. (1996), *Multiple Comparisons: Theory and Methods*, London: Chapman & Hall.
- Hsu, J. C. and Peruggia, M. (1994), "Graphical Representation of Tukey's Multiple Comparison Method," *Journal of Computational and Graphical Statistics*, 3, 143–161.
- Rodriguez, R. N. (2004), "An Introduction to ODS for Statistical Graphics in SAS 9.1," in *Proceedings of the Twenty-ninth Annual SAS Users Group International Conference*, Cary, NC: SAS Institute, Inc.
- Rodriguez, R. N. and Balan, T. E. (2006), "Creating Statistical Graphics in SAS 9.2: What Every Statistical User Should Know," in *Proceedings of the Thirty-first Annual SAS Users Group International Conference*, Cary, NC: SAS Institute Inc.
- SAS Institute Inc. (2008a), *Base SAS 9.2 Procedures Guide: Statistical Procedures*, Cary, NC: SAS Institute Inc.
- SAS Institute Inc. (2008b), *SAS/ETS 9.2 User's Guide*, Cary, NC: SAS Institute Inc.
- SAS Institute Inc. (2008c), *SAS/GRAPH 9.2: Graph Template Language Reference*, Cary, NC: SAS Institute Inc.
- SAS Institute Inc. (2008d), *SAS/QC 9.2 User's Guide*, Cary, NC: SAS Institute Inc.
- SAS Institute Inc. (2008e), *SAS/STAT 9.2 User's Guide*, Cary, NC: SAS Institute Inc.

## ACKNOWLEDGEMENTS

I am grateful to Jeff Cartier, Anne Jones, Warren Kuhfeld, Ann Kuo, and Ying So for valuable assistance in the preparation of this paper.

## RECOMMENDED READING

To get started using ODS Graphics with one of the statistical procedures listed on page 10, begin with the examples and the “Syntax” section of the procedure chapter in the user’s guides for SAS/STAT, SAS/ETS, SAS/QC, and the statistical procedures in Base SAS. Each chapter provides a section titled “ODS Graphics” that lists the available graphs. To get started with the new statistical graphics procedures in SAS/GRAPH, see the *SAS/GRAPH Statistical Graphics Procedures Guide*. Chapter 21, “Statistical Graphics Using ODS,” of the *SAS/STAT 9.2 User’s Guide* covers the general functionality of ODS Graphics.

## CONTACT INFORMATION

Robert N. Rodriguez  
SAS Institute Inc.  
SAS Campus Drive  
Cary, NC 27513  
(919) 531-7650  
Bob.Rodriguez@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.